

Solutions to the Exercises

EXERCISE 3

```
beginfig(1);
draw fullcircle scaled 2cm;
endfig;
end;
```

EXERCISE 4

```
beginfig(1);
path p;
p := (0,0)--(2cm,0)--(1cm,sqrt(3)*cm)--(0,0);
draw p;
endfig;
```

```
beginfig(2);
draw p scaled 1.5;
endfig;
```

```
end;
```

EXERCISE 5

```
beginfig(1);
u := 0.5cm;
draw (2u,0)--(u,sqrt(3)*u)--(-u,sqrt(3)*u)--(-2u,0)
--(-u,-sqrt(3)*u)--(u,-sqrt(3)*u)--(2u,0);
endfig;
end;
```

EXERCISE 6

```
beginfig(1);
u=1cm;
draw (0,0)--(2*sqrt(3)*u,0)--(sqrt(3)*u,3u)--(0,0);
draw (0,0)--(sqrt(3)*u,u)--(2*sqrt(3)*u,0);
draw (sqrt(3)*u,u)--(sqrt(3)*u,3u);
endfig;
```

```
beginfig(2);
draw unitsquare scaled 2u shifted (-u,-u);
draw unitsquare scaled 4u shifted (-2u,-2u);
draw (u,u)--(2u,2u);
draw (-u,u)--(-2u,2u);
draw (-u,-u)--(-2u,-2u);
draw (u,-u)--(2u,-2u);
endfig;
```

```
end;
```

EXERCISE 7

```
warningcheck := 0;
numeric p, q, n;
n := 12;
p := 2**n;
q := 2**n+1;
show p,q;
end;
```

EXERCISE 9

```
beginfig(1)
draw unitsquare scaled 70;
draw (10,20);
```

```
draw (10,15) scaled 2;
draw (30,40) withpen pencircle scaled 4;
pickup pencircle scaled 8;
draw (40,50);
draw (50,60);
endfig;
end;
```

EXERCISE 10

```
beginfig(1)
pickup pencircle scaled 6bp;
z.P = (1cm,2cm);
draw z.P;
draw 2(x.P,y.P);
endfig;
```

```
end;
```

EXERCISE 11

```
pair A,B,C,A',B',C';
u := 1cm;
A=(0,0);
B=(5u,0);
C=(2u,3u);
A'=1/2[B,C];
B'=1/2[A,C];
C'=1/2[A,B];
```

```
beginfig(1)
draw A--B--C--A;
draw A--A';
draw B--B';
draw C--C';
endfig;
```

```
beginfig(2)
draw A--B--C--A;
draw A--A';
draw B--B';
draw C--C';
dotlabel.lft("A",A);
dotlabel.urft("B",B);
dotlabel.top("C",C);
dotlabel.urft("A'",A');
dotlabel.ulft("B'",B');
dotlabel.bot("C'",C');
endfig;
```

```
beginfig(3)
pair G;
G = whatever[A,A'] = whatever[B,B'];
draw A--B--C--A;
draw A--A';
draw B--B';
draw C--C';
dotlabel.lft("A",A);
dotlabel.urft("B",B);
dotlabel.top("C",C);
dotlabel.urft("A'",A');
dotlabel.ulft("B'",B');
dotlabel.bot("C'",C');
dotlabel.llft("G",G);
label.llft("G",G-(0,1.5mm));
endfig;
end;
```

EXERCISE 12

```

beginfig(1)
s := 2cm;
z0 = s*dir(0);
z1 = s*dir(72);
z2 = s*dir(2*72);
z3 = s*dir(3*72);
z4 = s*dir(4*72);
draw z0--z1--z2--z3--z4--z0;
endfig;
end;

```

EXERCISE 13

```

beginfig(1);
z = (1cm,1cm);
draw z withpen pencircle scaled 6;
z1 = z - 2cm*dir(135);
z2 = z + 2cm*dir(135);
z3 = z + 2cm*dir(105);
draw z1--z2;
draw z--z3;
endfig;
end;

```

EXERCISE 14

```

pair p[]; p0 = (0,0); p1 = (2cm,3cm); p2 = (3cm,2cm);

beginfig(1);
fill p0--p1--p2--p0;
endfig;

beginfig(2);
fill p0--p1--p2--cycle withcolor 0.5white;
endfig;

end;

```

EXERCISE 15

```

beginfig(1);
draw origin--2*cm*dir(0);
draw origin--2*cm*dir(40);
drawarrow 1cm*dir(0){dir(90)}...
          1cm*dir(40){dir(130)}
endfig;

end;

```

EXERCISE 16

```

beginfig(1);
u := 1cm;
z0 = origin; z1 = (2u,0); z2 = (u,sqrt(3)*u);
draw z0--z1--z2;
draw z0{up}...z2;
draw z2{down}...z0;
pickup pencircle scaled 6;
draw z0; draw z1; draw z2;
endfig;

end;

```

EXERCISE 17

```

beginfig(1)
path p;
p = (0,1cm)..(1cm,0)..(0,-1cm);
fill p{dir(157)}..(0,0){dir(23)}..{dir(157)}cycle;
draw p..(-1cm,0)..cycle;
fill (0,-0.6cm)..(0.1cm,-0.5cm)..(0,-0.4cm)..
      (-0.1cm,-0.5cm)..cycle withcolor white;
fill (0,0.6cm)..(0.1cm,0.5cm)..(0,0.4cm)..(-0.1cm,0.5cm)..
      cycle;
endfig;
end;

```

EXERCISE 18

```

pair A,B,C,C';
path arc,mark[];
numeric AC, BC; % directional angle of AC and BC
u := 0.75cm; A=(0,0); B=(5u,0); C=(2u,3u);
AC = angle(A-C); BC = angle(B-C);
C' = whatever[A,B] = C + whatever*dir(1/2*AC+1/2*BC);
arc = (C+0.5u*dir(AC)){dir(AC+90)}..
      {dir(BC+90)}(C+0.5u*dir(BC));
mark[1] = C+0.4u*dir(3/4*AC+1/4*BC)--
          C+0.6u*dir(3/4*AC+1/4*BC);
mark[2] = C+0.4u*dir(1/4*AC+3/4*BC)--
          C+0.6u*dir(1/4*AC+3/4*BC);

beginfig(1)
draw A--B--C--cycle; draw C--C';
dotlabel.lft("A",A); dotlabel.urt("B",B);
dotlabel.top("C",C); dotlabel.bot("C'",C');
draw arc; draw mark[1]; draw mark[2];
endfig;
end;

```

EXERCISE 19

```

pair A,B,C,A',B',C',I;
u := 0.75cm; A=(0,0); B=(5u,-u); C=(2u,3u);
A' = whatever[B,C] = A + whatever*dir(
      1/2*angle(B-A)+1/2*angle(C-A));
B' = whatever[A,C] = B + whatever*dir(
      1/2*angle(A-B)+1/2*angle(C-B));
C' = whatever[A,B] = C + whatever*dir(
      1/2*angle(A-C)+1/2*angle(B-C));
I = whatever[A,A']=whatever[B,B'];

beginfig(1)
draw A--B--C--cycle;
draw A--A'; draw B--B'; draw C--C';
draw A'..B'..C'..cycle;
dotlabel.lft("A",A); dotlabel.rt("B",B);
dotlabel.top("C",C); dotlabel.urt("A'",A');
dotlabel.ulft("B'",B'); dotlabel.bot("C'",C');
labeloffset := 0.3cm;
dotlabel.llft("I",I);
endfig;
end;

```

EXERCISE 20

```

verbatimtex
%&latex
\documentclass{article}
\begin{document}
etex

```

```

beginfig(1)
u := 1cm;
numeric xmin, xmax, ymin, ymax, xinc;
xmin := 0; xmax := 4;
ymin := 0; ymax := 2;
% draw axes
draw (xmin,0)*u -- (xmax,0)*u;
draw (0,ymin)*u -- (0,ymax)*u;
% compute and draw graph of function
path p;
xinc := 0.1;
p := (xmin,sqrt(xmin))*u
  for x=xmin+xinc step xinc until xmax:
    .. (x,sqrt(x))*u
  endfor;
draw p withpen pencircle scaled 2;
% draw tickmarks and labels
for i=0 upto xmax:
  label.bot(decimal(i), (i,0)*u);
  draw (i,-0.05)*u--(i,0.05)*u;
endfor;
for i=0 upto ymax:
  label.lft(decimal(i), (0,i)*u);
  draw (-0.05,i)*u--(0.05,i)*u;
endfor;
\labeloffset := 0.5u;
label.bot(btex  $x$  etex, ((xmin+xmax)/2,0)*u);
label.lft(btex  $y$  etex, (0,(ymin+ymax)/2)*u);
label(btex  $y=\sqrt{x}$  etex,
  ((xmin+3xmax)/4, (ymin+2ymax)/3)*u);
endfig;
end;

```

EXERCISE 21

```

verbatimtex
%&latex
\documentclass{article}
\usepackage{amsmath,amssymb}
\begin{document}
etex

beginfig(1)
a := 3cm; b := 2cm;
phi := angle(a,b);
draw (-1/2cm,0)--(a+1/2cm,0); % horizontal axis
draw (0,-1/2cm)--(0,b+1/2cm); % vertical axis
draw (a,0)--(a,b)--(0,b) dashed evenly;
draw origin--(a,b);
label.llft(btex  $0$  etex, (0,0));
label.bot(btex  $a$  etex, (a,0));
label.lft(btex  $b$  etex, (0,b));
label.lft(btex  $\mathbb{C}$  etex, (0,b+1/2cm));
label.rt(btex  $a+ib=z$  etex, (a,b));
label(btex  $\lvert z \rvert$  etex, 1/2(a,b) + dir(90+phi)*1.5mm);
label(btex  $\lvert z \rvert$  etex, 1/2(a,b) + dir(90+phi)*5mm);
draw (1/2cm,0){up}.1/2cm*dir(phi){dir(90+phi)};
label(btex  $\phi$  etex, 7mm*dir(phi/2));
endfig;
end;

```

EXERCISE 22

```

verbatimtex
%&latex
\documentclass{article}

```

```

\begin{document}
etex

beginfig(1)
u := 0.6cm;
labeloffset := 1/3u;
defaultscale := 8pt/fontsize(defaultfont);
% the data
z1 = (1980,86); z2 = (1985,85); z3 = (1990,91);
z4 = (1995,86); z5 = (2000,83);
yoff := 80; % vertical offset
% draw axes
draw (0,0)--(1/4u,0)--(1/3u,1/8u)--(5/12u,-1/8u)
  --(1/2u,0)--(6u,0);
draw (0,0)--(0,1/4u)--(-1/8u,1/3u)--(1/8u,5/12u)
  --(0,1/2u)--(0,6.5u);
% draw horizontal axis and data points
for i=1 upto 5:
  draw (1/2+i,-1/12)*u--(1/2+i,1/12)*u; % ticks
  label.bot(decimal(x[i]), ((1/2+i)*u,0)); % labels
  dotlabel("", (1/2+i,(y[i]-yoff)/2)*u); % data point
endfor
% draw line graph
draw (3/2,(y[1]-yoff)/2)*u
  for i=2 upto 5: --(1/2+i, (y[i]-yoff)/2)*u endfor;
% draw vertical axis
for i=1 upto 6:
  draw (-1/12,i)*u--(1/12,i)*u; % ticks
  label.lft(decimal(yoff+2i), (0,i*u)); % labels
endfor
% draw horizontal and vertical texts
label.bot(btex year etex, (7/2u,-1/2u));
label.lft(btex beer consumption (liter) etex
  rotated 90, (-1/2u,7/2u));
endfig;

end;

```

EXERCISE 23

```

u := 1/2cm; defaultscale := 8pt/fontsize(defaultfont);
beginfig(1);
path sqr; sqr := unitsquare scaled u;
for i=0 upto 10:
  label.bot(decimal(i/10), ((i+1/2)*u,0));
  label.lft(decimal(i/10), (0,(i+1/2)*u));
  for j=0 upto 10:
    fill sqr shifted (i*u,j*u) withcolor
      (i*0.1+j*0.1)/3*white;
    draw sqr shifted (i*u,j*u); % for drawing the grid
  endfor;
endfor;
label.bot("r", (6u,-2/3u));
label.lft("g", (-u,6u));
label.top("RGB(r,g,0)", (6u,11u));
endfig;

beginfig(2);
path sqr; sqr := unitsquare scaled u;
for i=0 upto 10:
  label.bot(decimal(i/10), ((i+1/2)*u,0));
  label.lft(decimal(i/10), (0,(i+1/2)*u));
  for j=0 upto 10:
    fill sqr shifted (i*u,j*u) withcolor
      (0.3*i*0.1+0.59*j*0.1)*white;
    draw sqr shifted (i*u,j*u); % for drawing the grid
  endfor;
endfor;

```

```

endfor;
label.bot("r", (6u,-2/3u));
label.lft("g", (-u,6u));
label.top("RGB(r,g,0)", (6u,11u));
endfig;

```

```
end;
```

EXERCISE 24

```

beginfig(1);
path O[], l[]; pair A[];
O1 = fullcircle xscaled 1cm yscaled 1/2cm shifted (0,1cm);
O2 = O1 rotated -120;
A1 = (-1/2cm,1cm);
A2 = A1 xscaled -1;
l1 = A1{down}..(A2 rotated -120){down rotated 60};
l2 = A2{down}..(A1 rotated -120){down rotated 60};
draw O1; draw O2; draw l1; draw l2;
endfig;
end;

```

EXERCISE 25

```

beginfig(1);
r := 3cm;
path C[], p[], a;
C1 = fullcircle scaled 2r;
C2 = fullcircle scaled (2/3*2r);
p1 = origin -- r*dir(20);
p2 = origin -- r*dir(35);
a := buildcycle(p1,C2,p2,C1);
fill a withcolor red+green;
draw a withpen pencircle scaled 1bp;
draw p1; draw p2;
endfig;
end;

```

EXERCISE 26

```

beginfig(1);
u := 1cm; a := 6u; b := 3.5u;
pair sun; sun := (-1.75u,0);
path E[], p[], area[];
E1 = fullcircle xscaled a yscaled b;
E2 = E1 scaled 1.1;
p1 = sun -- (5u*dir(8) shifted sun);
p2 = sun -- (4u*dir(28) shifted sun);
p3 = sun -- (2u*dir(75) shifted sun);
p4 = sun -- (1.75u*dir(150) shifted sun);
area1 = buildcycle(p1,E1,p2);
area2 = buildcycle(p3,E1,p4);
fill area1 withcolor red+green;
fill area2 withcolor red+green;
draw p1; draw p2; draw p3; draw p4; draw E1;
draw (-a/2,0)--(a/2,0) dashed withdots;
draw sun withpen pencircle scaled 6bp;
label.bot(btex Sun etex, sun);
numeric t[]; % intersection times
t1 = ypart (p1 intersectiontimes E2);
t2 = ypart (p2 intersectiontimes E2);
t3 = ypart (p3 intersectiontimes E2);
t4 = ypart (p4 intersectiontimes E2);
drawarrow subpath (t1,t2) of E2;
drawarrow subpath (t3,t4) of E2;
label.urt(btex $\Delta t$ etex, point((t1+t2)/2) of E2);

```

```

label.ulft(btex $\Delta t'$ etex, point((t3+t4)/2) of E2);
endfig;

```

```
end;
```

EXERCISE 27

```

beginfig(1);
u:=1cm;
draw (-2u,0)--(2u,0);
draw (0,-2u)--(0,2u);
for i=-2u step u until 2u:
  draw (i,u/10)--(i,-u/10);
  draw (u/10,i)--(-u/10,i);
endfor;
for i=-2u step u/5 until 2u:
  draw (i,u/20)--(i,-u/20);
  draw (u/20,i)--(-u/20,i);
endfor;
endfig;
end;

```

EXERCISE 28

```

beginfig(1);
u:=1cm;
numeric xmin, xmax, ymin, ymax;
xmin := -2.1; xmax := 2.1;
ymin := -0.5; ymax := 4.5;
% draw axes
path xaxis, yaxis;
xaxis = (xmin,0)*u -- (xmax,0)*u;
yaxis = (0,ymin)*u -- (0,ymax)*u;
% compute the graph of f
def f(expr x) = (4-x**2) enddef;
inc := 0.01;
path pts_f;
pts_f := (xmin*u,f(xmin)*u)
  for x=xmin+inc step inc until xmax:
    .. (x*u,f(x)*u)
  endfor;
% compute and draw rectangles
n := 12; % number of rectangles
x0 := -2; x1 := 2;
inc := (x1-x0)/n;
for i=x0 step inc until x1-inc:
  path p;
  p = (i,0)--(i+inc,0)--(i+inc,max(f(i),f(i+inc)))
    --(i, max(f(i),f(i+inc)))--cycle;
  p := p scaled u;
  fill p withcolor red+green;
  draw p;
endfor;
draw pts_f withpen pencircle scaled 2;
draw xaxis;
draw yaxis;
pair t; % translation vector
t := (6u,0);
for i=x0 step inc until x1-inc:
  path p;
  p = (i,0)--(i+inc,0)--(i+inc,min(f(i),f(i+inc)))
    --(i, min(f(i),f(i+inc)))--cycle;
  p := p scaled u shifted t;
  fill p withcolor red+green;
  draw p;
endfor;
draw pts_f shifted t withpen pencircle scaled 2;

```

```
draw xaxis shifted t;
draw yaxis shifted t;
endfig;
```

```
end;
```

EXERCISE 29

```
beginfig(1)
u:=1cm;
draw (-2u,0)--(2u,0);
draw (0,-2u)--(0,2u);
for i=-2u step u until 2u:
  draw(i,2u)--(i,-2u);
  draw (2u,i)--(-2u,i);
endfor;
for i=-2u step u/10 until 2u:
  draw (i,2u)--(i,-2u) withpen pencircle scaled .1bp;
  draw (2u,i)--(-2u,i) withpen pencircle scaled .1bp;
endfor;
endfig;
end;
```

EXERCISE 30

```
beginfig(1)
u:=1cm;
pair A[], B[];
n := 3;
for i=1 upto n:
  A[i] = (0,i*u);
  B[i] = (n/2*u,i*u);
endfor;
for i=1 upto n:
  for j=1 upto n:
    draw A[i]--B[j];
  endfor;
endfor;
for i=1 upto n:
  dotlabel.lft("a" & decimal(i), A[i]);
  dotlabel.rt("b" & decimal(i), B[i]);
endfor;
endfig;
end;
```

EXERCISE 31

```
u:=3cm;
vardef koch(expr A,B,n) =
  save C; pair C;
```

```
  C = A rotatedaround(1/3[A,B], 120);
  if n>1:
    koch( A,      1/3[A,B], n-1);
    koch( 1/3[A,B], C,      n-1);
    koch( C,      2/3[A,B], n-1);
    koch( 2/3[A,B], B,      n-1);
  else:
    draw A--1/3[A,B]--C--2/3[A,B]--B;
  fi;
enddef;
```

```
beginfig(1)
z0=(u,0);
z1=z0 rotated 120;
z2=z1 rotated 120;
draw z0--z1--z2--cycle shifted (-3u,0);
drawarrow (-1.75u,0)--(-1.25u,0);
koch( z0, z1, 1);
koch( z1, z2, 1);
koch( z2, z0, 1);
endfig;
```

```
beginfig(2)
z0=(u,0);
z1=z0 rotated 120;
z2=z1 rotated 120;
koch( z0, z1, 2);
koch( z1, z2, 2);
koch( z2, z0, 2);
endfig;
```

```
beginfig(3)
z0=(u,0);
z1=z0 rotated 120;
z2=z1 rotated 120;
koch( z0, z1, 6);
koch( z1, z2, 6);
koch( z2, z0, 6);
endfig;
```

```
end;
```

André Heck
© 2003, AMSTEL Institute,
Universiteit van Amsterdam